

## GRAMATIFA EN BNF

<Programa> ::= "program" "id" ";" <Definiciones> "{" <Cuerpo> "}"

<Definiciones> ::= "def" <ListaDefiniciones> | eps

<ListaDefiniciones> ::= <Definicion> ";" <ListaDefiniciones> | <Definicion> ";"

<Definicion> ::= "id" ":" <Tipo>

<Tipo> ::= "matriz" "[" <OP> "]" "[" <OP> "]" | "real"

<Cuerpo> ::= <Sentencias> ";" <Cuerpo> | eps

<Sentencias> ::= <Asignacion> | <Leer> | <Escribir> | <Condicional> | <Ciclo>

<Asignacion> ::= "id" ":" <OP> | "id" ":" <CMatriz> | "id" "[" <OP> "]" "[" <OP> "]" ":" <OP>

<OP> ::= <OP> "+" <OP2> | <OP> "-" <OP2> | <OP2>

<OP2> ::= <OP2> "\*" <OP3> | <OP2> "/" <OP3> | <OP3>

<OP3> ::= <OP3> "^" <OP4> | <OP4>

<OP4> ::= "id" "[" <OP> "]" "[" <OP> "]" | "id" | "creal" | "filas" "(" "id" ")" | "columnas" "(" "id" ")"  
| "-" <OP4> | "tras" "(" "id" ")" | "(" <OP> ")"

<CMatriz> ::= "[" <Filas> "]"

<Filas> ::= <Fila> | <Fila> , <Filas>

<Fila> ::= "[" <Numeros> "]"

<Numeros> ::= <OP4> | <OP4> ";" <Numeros>

<Leer> ::= "leer" "(" "cadena" , "id" ")"

<Escribir> ::= "escribir" "(" <Lista> ")"

<Lista> ::= <Elemento> | <Elemento> ";" <Lista>

<Elemento> ::= "cadena" | <OP>

<Condicional> ::= "if" <Condicion> "{" <Cuerpo> "}" <SiNo>

<SiNo> ::= "else" "{" <Cuerpo> "}" | eps

<Ciclo> ::= "while" <Condicion> "{" <Cuerpo> "}"

<Condicion> ::= "(" <ExpresionL> ")"

<ExpresionL> ::= <ExpresionL> "&&" <ExpresionR>

| <ExpresionL> "||" <ExpresionR>

| "!" "(" <ExpresionL> ")"

| <ExpresionR>

<ExpresionR> ::= <OP> "==" <OP>

| <OP> "!=" <OP>  
 | <OP> ">" <OP>  
 | <OP> "<" <OP>  
 | <OP> ">=" <OP>  
 | <OP> "<=" <OP>  
 | "(" <ExpresionL> ")"

## GRAMATICA LL1

<Programa> ::= "program" "id" ";" <Definiciones> "{" <Cuerpo> "}"

<Definiciones> ::= "def" <ListaDefiniciones> | eps

<ListaDefiniciones> ::= <Definicion> ";" <MasDefiniciones>

<MasDefiniciones> ::= <ListaDefiniciones> | eps

<Definicion> ::= "id" ":" <Tipo>

<Tipo> ::= "matriz" "[" "creal" "]" "[" "creal" "]" | "real"

<Cuerpo> ::= <Sentencias> ";" <Cuerpo> | eps

<Sentencias> ::= <Asignacion> | <Leer> | <Escribir> | <Condicional> | <Ciclo>

<Asignacion> ::= "id" <Asignacion'>

<Asignacion'> ::= ":" <Asignacion''> | "[" <OP> "]" "[" <OP> "]" ":" <OP>

<Asignacion''> ::= <OP> | <CMatriz>

<OP> ::= <OP2> <OP'>

<OP'> ::= "+" <OP2> <OP'> | "-" <OP2> <OP'> | eps

<OP2> ::= <OP3> <OP2'>

<OP2'> ::= "\*" <OP3> <OP2'> | "/" <OP3> <OP2'> | eps

<OP3> ::= <OP4> <OP3'>

<OP3'> ::= "^" <OP4> <OP3'> | eps

<OP4> ::= "id" <OP4'> | "creal" | "filas" "(" "id" ")" | "columnas" "(" "id" ")" |  
 "tras" "(" "id" ")" | "-" <OP4> | "(" <OP> ")"

<OP4'> ::= "[" <OP> "]" "[" <OP> "]" | eps

<CMatriz> ::= "[" <Filas> "]"

<Filas> ::= <Fila> <FilasExtra>

<FilasExtra> ::= "," <Filas> | eps

<Fila> ::= “[ <Numeros> ”]

<Numeros> ::= <OP4> <Numeros'>

<Numeros'> ::= “,” <Numeros> | ε

<Leer> ::= “leer” (“cadena” “,” “id” “”)

<Escribir> ::= “escribir” (“<Lista> ”)

<Lista> ::= <Elemento> <Lista'>

<Lista'> ::= “,” <Lista> | ε

<Elemento> ::= “cadena” | <OP>

<Condicional> ::= “if” <Condicion> “{” <Cuerpo> “}” <SiNo>

<SiNo> ::= “else” “{” <Cuerpo> “}” | eps

<Ciclo> ::= “while” <Condicion> “{” <Cuerpo> “}”

<Condicion> ::= “(” <ExpresionL> “)”

<ExpresionL> ::= <ExpresionR> <ExpresionL'>

| “!” “(” <ExpresionL> “)”

<ExpresionL'> ::= “&&” <ExpresionR> <ExpresionL'>

| “||” <ExpresionR> <ExpresionL'>

| eps

<ExpresionR> ::= “?” <OP> <Comparacion> <OP> “?” | “(” <ExpresionL> “)”

<Comparacion> ::= “==” | “!=” | “>” | “<” | “>=” | “<=”